

Higher order perturbation of the metric tensor

Preliminaries

Load tensor package

```
In[*]:= << xAct`xPert`  
-----  
Package xAct`xPerm` version 1.2.3, {2015, 8, 23}  
Copyright (C) 2003-2020, Jose M. Martin-Garcia, under the General Public License.  
Connecting to external linux executable...  
Connection established.  
-----  
Package xAct`xTensor` version 1.1.4, {2020, 2, 16}  
Copyright (C) 2002-2020, Jose M. Martin-Garcia, under the General Public License.  
-----  
Package xAct`xPert` version 1.0.6, {2018, 2, 28}  
Copyright (C) 2005-2018, David Brizuela, Jose M. Martin-Garcia and Guillermo A. Mena Marugan, under the General Public License.  
-----  
These packages come with ABSOLUTELY NO WARRANTY; for details type Disclaimer[]. This  
is free software, and you are welcome to redistribute it under certain conditions. See the General Public License for details.  
-----  
** Variable $PrePrint assigned value ScreenDollarIndices  
** Variable $CovDFormat changed from Prefix to Postfix  
** Option AllowUpperDerivatives of ContractMetric changed from False to True  
** Option MetricOn of MakeRule changed from None to All  
** Option ContractMetrics of MakeRule changed from False to True
```

Nicer printing

```
In[*]:= $PrePrint = ScreenDollarIndices;  
In[*]:= $CovDFormat = "Prefix";
```

Object definitions

Spacetime manifold

The spacetime manifold M , on which tensors will be defined. Some Greek letters are defined as tangent space indices.

```
In[ ]:= DefManifold[M, 4, { $\alpha$ ,  $\beta$ ,  $\gamma$ ,  $\mu$ ,  $\nu$ ,  $\rho$ ,  $\sigma$ ,  $\tau$ ,  $\kappa$ ,  $\xi$ ,  $\zeta$ }
** DefManifold: Defining manifold M.
** DefVBundle: Defining vbundle TangentM.
```

Metric

The metric g of signature $(-,+,+,+)$. The Levi-Civita derivative of a tensor A_μ will be written as $\nabla_\mu A_\nu$ in prefix notation or $A_{\nu;\mu}$ in postfix notation.

```
In[ ]:= DefMetric[{3, 1, 0}, Met[- $\mu$ , - $\nu$ ], CD, {";", "\nabla"}, PrintAs  $\rightarrow$  "g"]
** DefTensor: Defining symmetric metric tensor Met[- $\mu$ , - $\nu$ ].
** DefTensor: Defining antisymmetric tensor epsilonMet[- $\alpha$ , - $\beta$ , - $\gamma$ , - $\zeta$ ].
** DefTensor: Defining tetrametric TetraMet[- $\alpha$ , - $\beta$ , - $\gamma$ , - $\zeta$ ].
** DefTensor: Defining tetrametric TetraMet[- $\alpha$ , - $\beta$ , - $\gamma$ , - $\zeta$ ].
** DefCovD: Defining covariant derivative CD[- $\mu$ ].
** DefTensor: Defining vanishing torsion tensor TorsionCD[ $\alpha$ , - $\beta$ , - $\gamma$ ].
** DefTensor: Defining symmetric Christoffel tensor ChristoffelCD[ $\alpha$ , - $\beta$ , - $\gamma$ ].
** DefTensor: Defining Riemann tensor RiemannCD[- $\alpha$ , - $\beta$ , - $\gamma$ , - $\zeta$ ].
** DefTensor: Defining symmetric Ricci tensor RicciCD[- $\alpha$ , - $\beta$ ].
** DefCovD: Contractions of Riemann automatically replaced by Ricci.
** DefTensor: Defining Ricci scalar RicciScalarCD[].
** DefCovD: Contractions of Ricci automatically replaced by RicciScalar.
** DefTensor: Defining symmetric Einstein tensor EinsteinCD[- $\alpha$ , - $\beta$ ].
** DefTensor: Defining Weyl tensor WeylCD[- $\alpha$ , - $\beta$ , - $\gamma$ , - $\zeta$ ].
** DefTensor: Defining symmetric TFRicci tensor TFRicciCD[- $\alpha$ , - $\beta$ ].
** DefTensor: Defining Kretschmann scalar KretschmannCD[].
** DefCovD: Computing RiemannToWeylRules for dim 4
** DefCovD: Computing RicciToTFRicci for dim 4
** DefCovD: Computing RicciToEinsteinRules for dim 4
** DefTensor: Defining weight +2 density DetMet[]. Determinant.
```

Perturbations of the metric are written as δg .

```
In[ ]:= DefMetricPerturbation[Met,  $\delta$ Met,  $\epsilon$ , PrintAs  $\rightarrow$  " $\delta g$ "]
```

** DefParameter: Defining parameter ϵ .

** DefTensor: Defining tensor $\delta\text{Met}[\text{LI}[\text{order}], -\alpha, -\beta]$.

Taylor series expansion

Definition

Before studying perturbations of the metric tensor and other tensor derived from the metric, it is helpful to briefly recall the Taylor expansion of a function f around some value x :

In[]:= Series[f[x + δx], { δx , 0, 3}]

Out[]:=

$$f[x] + f'[x] \delta x + \frac{1}{2} f''[x] \delta x^2 + \frac{1}{6} f^{(3)}[x] \delta x^3 + O[\delta x]^4$$

Product rule

Since the linear perturbation is given by the derivative, the product rule holds:

In[]:= Series[f[x + δx] * g[x + δx], { δx , 0, 3}]

Out[]:=

$$f[x] * g[x] + (g[x] f'[x] + f[x] g'[x]) \delta x + \frac{1}{2} (2 f'[x] g'[x] + g[x] f''[x] + f[x] g''[x]) \delta x^2 + \frac{1}{6} (3 g'[x] f''[x] + 3 f'[x] g''[x] + g[x] f^{(3)}[x] + f[x] g^{(3)}[x]) \delta x^3 + O[\delta x]^4$$

Inverse rule

Similarly to the product rule, we have the rule

In[]:= Series[1 / f[x + δx], { δx , 0, 3}]

Out[]:=

$$\frac{1}{f[x]} - \frac{f'[x] \delta x}{f[x]^2} + \frac{(2 f'[x]^2 - f[x] f''[x]) \delta x^2}{2 f[x]^3} + \frac{(-6 f'[x]^3 + 6 f[x] f'[x] f''[x] - f[x]^2 f^{(3)}[x]) \delta x^3}{6 f[x]^4} + O[\delta x]^4$$

Notational conventions

Perturbation of the metric tensor

The metric is expanded in a formal Taylor series expansion, where the n 'th perturbation order carries a factor $\frac{\epsilon^n}{n!}$ and is written as $\delta g_{\mu\nu}^n$.

In[]:= Perturbed[Met[- μ , - ν], 3]

Out[]:=

$$g_{\mu\nu} + \epsilon \delta g_{\mu\nu}^1 + \frac{1}{2} \epsilon^2 \delta g_{\mu\nu}^2 + \frac{1}{6} \epsilon^3 \delta g_{\mu\nu}^3$$

This gives us a particular perturbation order.

```
In[*]:= Perturbation[Met[-μ, -ν], 2]
```

```
Out[*]=
```

$$\delta g^2_{\mu\nu}$$

We can also obtain it as a series expansion coefficient, given by the derivative.

```
In[*]:= D[Perturbed[Met[-μ, -ν], 3], {ε, 2}] /. ε → 0
```

```
Out[*]=
```

$$\delta g^2_{\mu\nu}$$

Perturbation of other tensors

For other tensors, the n'th order perturbation is written as $\Delta^n[X]$:

```
In[*]:= Perturbation[RicciCD[-μ, -ν], 2]
```

```
Out[*]=
```

$$\Delta^0[R[\nabla]_{\mu\nu}]$$

The formal series expansion has the same form:

```
In[*]:= Perturbed[RicciCD[-μ, -ν], 3]
```

```
Out[*]=
```

$$\epsilon \Delta[R[\nabla]_{\mu\nu}] + \frac{1}{2} \epsilon^2 \Delta^0[R[\nabla]_{\mu\nu}] + \frac{1}{6} \epsilon^3 \Delta^1[R[\nabla]_{\mu\nu}] + R[\nabla]_{\mu\nu}$$

Perturbation of the inverse metric

Relation with the Kronecker symbol

We start by calculating the perturbation $\Delta[g^{\mu\nu}]$ of inverse $g^{\mu\nu}$ of the metric tensor. This can be obtained from two steps. First, note the relation between the metric and its inverse, whose contraction yields a Kronecker symbol:

```
In[*]:= Inactive[Met[μ, ρ]] × Inactive[Met[-ρ, -ν]]
```

```
Activate[%]
```

```
Out[*]=
```

$$\text{Inactive}[g^{\mu\rho}] \times \text{Inactive}[g_{\rho\nu}]$$

```
Out[*]=
```

$$\delta_\nu^\mu$$

Second, since the Kronecker symbol does not depend on the metric, its perturbation at any positive order vanishes, which we see from the absence of any terms depending on ϵ :

```
In[*]:= Perturbed[delta[-μ, ν], 4]
```

```
Out[*]=
```

$$\delta_\mu^\nu$$

First order

We can use this to calculate the perturbation of the inverse metric. Here we use the product rule, starting with the first order, to obtain a rule one can use later:

```

In[ ]:= Perturbation[Inactive[Met[μ, ρ]] * Inactive[Met[-ρ, -σ]], 1];
Activate[%]
Expand[Met[v, σ] %]
Solve[% == 0, Perturbation[Met[μ, ν]][[1]]]
δimet1 = MakeRule[Evaluate[List @@ (%[[1]])]];

```

```
Out[ ]:=
```

$$g_{\rho\sigma} \Delta[g^{\mu\rho}] + g^{\mu\rho} \delta g^1_{\rho\sigma}$$

```
Out[ ]:=
```

$$\Delta[g^{\mu\nu}] + g^{\mu\rho} g^{\nu\sigma} \delta g^1_{\rho\sigma}$$

```
Out[ ]:=
```

$$\{\Delta[g^{\mu\nu}] \rightarrow -g^{\mu\rho} g^{\nu\sigma} \delta g^1_{\rho\sigma}\}$$

This perturbation can also easily be obtained from the built-in rules, and we see that the result is the same.

```

In[ ]:= Perturbation[Met[μ, ν], 1]
ExpandPerturbation[%]
SeparateMetric[[%]]

```

```
Out[ ]:=
```

$$\Delta[g^{\mu\nu}]$$

```
Out[ ]:=
```

$$- \delta g^{1\mu\nu}$$

```
Out[ ]:=
```

$$- g^{\mu\alpha} g^{\nu\beta} \delta g^1_{\alpha\beta}$$

Second order

With the same line of argument, we can derive the second perturbation order. Here we need the result from the first step.

```

In[ ]:= Perturbation[Inactive[Met[μ, ρ]] * Inactive[Met[-ρ, -σ]], 2];
Activate[%]
% /. δimet1
SeparateMetric[[][%];
Expand[Met[v, σ] %];
ToCanonical[%]
Solve[% == 0, Perturbation[Met[μ, v], 2]][[1]]
δimet2 = MakeRule[Evaluate[List @@ (%[[1]])]];

```

Out[]:=

$$g_{\rho\sigma} \Delta^0[g^{\mu\rho}] + 2 \Delta[g^{\mu\rho}] \delta g^1_{\rho\sigma} + g^{\mu\rho} \delta g^2_{\rho\sigma}$$

Out[]:=

$$g_{\rho\sigma} \Delta^0[g^{\mu\rho}] - 2 \delta g^{1\mu\rho} \delta g^1_{\rho\sigma} + g^{\mu\rho} \delta g^2_{\rho\sigma}$$

Out[]:=

$$\Delta^0[g^{\mu\nu}] - 2 g^{\mu\alpha} g^{\nu\beta} g^{\rho\sigma} \delta g^1_{\alpha\rho} \delta g^1_{\beta\sigma} + g^{\mu\alpha} g^{\nu\beta} \delta g^2_{\alpha\beta}$$

Out[]:=

$$\left\{ \Delta^0[g^{\mu\nu}] \rightarrow 2 g^{\mu\alpha} g^{\nu\beta} g^{\rho\sigma} \delta g^1_{\alpha\rho} \delta g^1_{\beta\sigma} - g^{\mu\alpha} g^{\nu\beta} \delta g^2_{\alpha\beta} \right\}$$

Again we verify our result.

```

In[ ]:= Perturbation[Met[μ, v], 2]
ExpandPerturbation[%]
SeparateMetric[[][%]

```

Out[]:=

$$\Delta^0[g^{\mu\nu}]$$

Out[]:=

$$2 \delta g^1_{\alpha}{}^{\nu} \delta g^{1\mu\alpha} - \delta g^{2\mu\nu}$$

Out[]:=

$$2 g^{\alpha\zeta} g^{\mu\gamma} g^{\nu\beta} \delta g^1_{\alpha\beta} \delta g^1_{\gamma\zeta} - g^{\mu\alpha} g^{\nu\beta} \delta g^2_{\alpha\beta}$$

General formula for higher orders

Now we aim for a general formula for arbitrary high perturbation orders. We know that the perturbation of the Kronecker symbol vanishes at all orders, and so the following terms vanish by definition.

```
In[*]:= Table[Perturbation[Inactive[Met[μ, ρ]] × Inactive[Met[-ρ, -σ]], k], {k, 4};
Activate[%];
Expand[Met[v, σ] %];
TableForm[%]
```

Out[*]//TableForm=

$$\begin{aligned} & \Delta[g^{\mu\nu}] + g^{\mu\rho} g^{\nu\sigma} \delta g^1_{\rho\sigma} \\ & \Delta^0[g^{\mu\nu}] + 2 g^{\nu\sigma} \Delta[g^{\mu\rho}] \delta g^1_{\rho\sigma} + g^{\mu\rho} g^{\nu\sigma} \delta g^2_{\rho\sigma} \\ & \Delta^1[g^{\mu\nu}] + 3 g^{\nu\sigma} \Delta^0[g^{\mu\rho}] \delta g^1_{\rho\sigma} + 3 g^{\nu\sigma} \Delta[g^{\mu\rho}] \delta g^2_{\rho\sigma} + g^{\mu\rho} g^{\nu\sigma} \delta g^3_{\rho\sigma} \\ & \Delta^2[g^{\mu\nu}] + 4 g^{\nu\sigma} \Delta^1[g^{\mu\rho}] \delta g^1_{\rho\sigma} + 6 g^{\nu\sigma} \Delta^0[g^{\mu\rho}] \delta g^2_{\rho\sigma} + 4 g^{\nu\sigma} \Delta[g^{\mu\rho}] \delta g^3_{\rho\sigma} + g^{\mu\rho} g^{\nu\sigma} \delta g^4_{\rho\sigma} \end{aligned}$$

We can obtain this expansion using the Taylor expansion in the perturbation parameter:

```
In[*]:= Table[D[Perturbed[Met[μ, ρ], k] × Perturbed[Met[-ρ, -σ], k], {ε, k}] /. ε → 0, {k, 4};
Expand[Met[v, σ] %];
TableForm[%]
```

Out[*]//TableForm=

$$\begin{aligned} & \Delta[g^{\mu\nu}] + g^{\mu\rho} g^{\nu\sigma} \delta g^1_{\rho\sigma} \\ & \Delta^0[g^{\mu\nu}] + 2 g^{\nu\sigma} \Delta[g^{\mu\rho}] \delta g^1_{\rho\sigma} + g^{\mu\rho} g^{\nu\sigma} \delta g^2_{\rho\sigma} \\ & \Delta^1[g^{\mu\nu}] + 3 g^{\nu\sigma} \Delta^0[g^{\mu\rho}] \delta g^1_{\rho\sigma} + 3 g^{\nu\sigma} \Delta[g^{\mu\rho}] \delta g^2_{\rho\sigma} + g^{\mu\rho} g^{\nu\sigma} \delta g^3_{\rho\sigma} \\ & \Delta^2[g^{\mu\nu}] + 4 g^{\nu\sigma} \Delta^1[g^{\mu\rho}] \delta g^1_{\rho\sigma} + 6 g^{\nu\sigma} \Delta^0[g^{\mu\rho}] \delta g^2_{\rho\sigma} + 4 g^{\nu\sigma} \Delta[g^{\mu\rho}] \delta g^3_{\rho\sigma} + g^{\mu\rho} g^{\nu\sigma} \delta g^4_{\rho\sigma} \end{aligned}$$

This can now be written explicitly as a sum over products of perturbations.

```
In[*]:= Table[Sum[Binomial[k, j] Perturbation[Met[μ, ρ], k - j] × Perturbation[Met[-ρ, -σ], j], {j, 0, k}], {k, 4};
Expand[Met[v, σ] %];
TableForm[%]
```

Out[*]//TableForm=

$$\begin{aligned} & \Delta[g^{\mu\nu}] + g^{\mu\rho} g^{\nu\sigma} \delta g^1_{\rho\sigma} \\ & \Delta^0[g^{\mu\nu}] + 2 g^{\nu\sigma} \Delta[g^{\mu\rho}] \delta g^1_{\rho\sigma} + g^{\mu\rho} g^{\nu\sigma} \delta g^2_{\rho\sigma} \\ & \Delta^1[g^{\mu\nu}] + 3 g^{\nu\sigma} \Delta^0[g^{\mu\rho}] \delta g^1_{\rho\sigma} + 3 g^{\nu\sigma} \Delta[g^{\mu\rho}] \delta g^2_{\rho\sigma} + g^{\mu\rho} g^{\nu\sigma} \delta g^3_{\rho\sigma} \\ & \Delta^2[g^{\mu\nu}] + 4 g^{\nu\sigma} \Delta^1[g^{\mu\rho}] \delta g^1_{\rho\sigma} + 6 g^{\nu\sigma} \Delta^0[g^{\mu\rho}] \delta g^2_{\rho\sigma} + 4 g^{\nu\sigma} \Delta[g^{\mu\rho}] \delta g^3_{\rho\sigma} + g^{\mu\rho} g^{\nu\sigma} \delta g^4_{\rho\sigma} \end{aligned}$$

Note that the first element in this sum is exactly the perturbation we try to solve for. Hence, we can write the orders as follows.

```
In[*]:= Table[Perturbation[Met[μ, ν], k] → -Expand[Met[ν, σ] × Sum[Binomial[k, j] Perturbation[Met[μ, ρ], k - j] × Perturbation[Met[-ρ, -σ], j], {j, k}]], {k, 4};
δimet1to4 = Flatten[MakeRule[Evaluate[List @@ #]] & /@ %];
TableForm[%]
```

Out[*]//TableForm=

$$\begin{aligned} \Delta[g^{\mu\nu}] &\rightarrow -g^{\mu\rho} g^{\nu\sigma} \delta g^1_{\rho\sigma} \\ \Delta^0[g^{\mu\nu}] &\rightarrow -2 g^{\nu\sigma} \Delta[g^{\mu\rho}] \delta g^1_{\rho\sigma} - g^{\mu\rho} g^{\nu\sigma} \delta g^2_{\rho\sigma} \\ \Delta^1[g^{\mu\nu}] &\rightarrow -3 g^{\nu\sigma} \Delta^0[g^{\mu\rho}] \delta g^1_{\rho\sigma} - 3 g^{\nu\sigma} \Delta[g^{\mu\rho}] \delta g^2_{\rho\sigma} - g^{\mu\rho} g^{\nu\sigma} \delta g^3_{\rho\sigma} \\ \Delta^2[g^{\mu\nu}] &\rightarrow -4 g^{\nu\sigma} \Delta^1[g^{\mu\rho}] \delta g^1_{\rho\sigma} - 6 g^{\nu\sigma} \Delta^0[g^{\mu\rho}] \delta g^2_{\rho\sigma} - 4 g^{\nu\sigma} \Delta[g^{\mu\rho}] \delta g^3_{\rho\sigma} - g^{\mu\rho} g^{\nu\sigma} \delta g^4_{\rho\sigma} \end{aligned}$$

To arrive at the final result, we still need to replace the lower orders which appear on the right hand side of these rules, by recursively applying the lower order rules. Here we contract the metrics, to obtain a more compact result.

```
In[*]:= Table[Perturbation[Met[μ, ν], k], {k, 4};
% /. δimet1to4;
Expand[%];
ContractMetric /@ %;
ToCanonical /@ %;
ScreenDollarIndices /@ %;
TableForm[%]
```

Out[*]//TableForm=

$$\begin{aligned} & -\delta g^{1\mu\nu} \\ & 2 \delta g^{1\mu\alpha} \delta g^{1\nu}_\alpha - \delta g^{2\mu\nu} \\ & -6 \delta g^1_{\alpha\beta} \delta g^{1\mu\alpha} \delta g^{1\nu\beta} + 3 \delta g^{1\nu}_\alpha \delta g^{2\mu\alpha} + 3 \delta g^{1\mu\alpha} \delta g^{2\nu}_\alpha - \delta g^{3\mu\nu} \\ & 24 \delta g^1_\alpha \delta g^1_{\beta\gamma} \delta g^{1\mu\alpha} \delta g^{1\nu\beta} - 12 \delta g^{1\mu\alpha} \delta g^{1\nu\beta} \delta g^2_{\alpha\beta} - 12 \delta g^1_{\alpha\beta} \delta g^{1\nu\beta} \delta g^{2\mu\alpha} + 6 \delta g^{2\mu\alpha} \delta g^{2\nu}_\alpha - 12 \delta g^1_{\alpha\beta} \delta g^{1\mu\alpha} \delta g^{2\nu\beta} + 4 \delta g^{1\nu}_\alpha \delta g^{3\mu\alpha} + 4 \delta g^{1\mu\alpha} \delta g^{3\nu}_\alpha - \delta g^{4\mu\nu} \end{aligned}$$

This knowledge allows us to define a general formula for arbitrary perturbation order.

```
In[*]:= δimet[0] := {}
In[*]:= δimet[k_Integer /; k > 0] := Module[{j, μ, ν, ρ, σ, ru}, ru = δimet[k - 1];
Join[ru, MakeRule[Evaluate[{Perturbation[Met[μ, ν], k], -Expand[Met[ν, σ] × Sum[Binomial[k, j] Perturbation[Met[μ, ρ], k - j] × Perturbation[Met[-ρ, -σ], j], {j, k}]] /. ru}]]]]]
```

This is what we obtain from the function just defined.

```
In[*]:= Table[Perturbation[Met[μ, ν], k] /. δimet[k], {k, 4};
Expand[%];
ContractMetric /@ %;
ToCanonical /@ %;
ScreenDollarIndices /@ %;
TableForm[%]
```

Out[*]//TableForm=

$$\begin{aligned} & -\delta g^{1\mu\nu} \\ & 2 \delta g^{1\mu\alpha} \delta g^{1\nu}_\alpha - \delta g^{2\mu\nu} \\ & -6 \delta g^1_{\alpha\beta} \delta g^{1\mu\alpha} \delta g^{1\nu\beta} + 3 \delta g^{1\nu}_\alpha \delta g^{2\mu\alpha} + 3 \delta g^{1\mu\alpha} \delta g^{2\nu}_\alpha - \delta g^{3\mu\nu} \\ & 24 \delta g^1_\alpha \delta g^1_{\beta\gamma} \delta g^{1\mu\alpha} \delta g^{1\nu\beta} - 12 \delta g^{1\mu\alpha} \delta g^{1\nu\beta} \delta g^2_{\alpha\beta} - 12 \delta g^1_{\alpha\beta} \delta g^{1\nu\beta} \delta g^{2\mu\alpha} + 6 \delta g^{2\mu\alpha} \delta g^{2\nu}_\alpha - 12 \delta g^1_{\alpha\beta} \delta g^{1\mu\alpha} \delta g^{2\nu\beta} + 4 \delta g^{1\nu}_\alpha \delta g^{3\mu\alpha} + 4 \delta g^{1\mu\alpha} \delta g^{3\nu}_\alpha - \delta g^{4\mu\nu} \end{aligned}$$

We check this result.

```
In[ ]:= Table[Perturbation[Met[μ, ν], k], {k, 4}];
ExpandPerturbation /@ %;
Expand[%];
ContractMetric /@ %;
ToCanonical /@ %;
ScreenDollarIndices /@ %;
TableForm[%]
```

Out[]//TableForm=

$$\begin{aligned}
 & -\delta g^{1\mu\nu} \\
 & 2\delta g^{1\mu\alpha}\delta g^{1\nu}_{\alpha} - \delta g^{2\mu\nu} \\
 & -6\delta g^{1\alpha\beta}\delta g^{1\mu\alpha}\delta g^{1\nu\beta} + 3\delta g^{1\nu}_{\alpha}\delta g^{2\mu\alpha} + 3\delta g^{1\mu\alpha}\delta g^{2\nu}_{\alpha} - \delta g^{3\mu\nu} \\
 & 24\delta g^{1\gamma}_{\alpha}\delta g^{1\beta\gamma}\delta g^{1\mu\alpha}\delta g^{1\nu\beta} - 12\delta g^{1\mu\alpha}\delta g^{1\nu\beta}\delta g^{2\alpha\beta} - 12\delta g^{1\alpha\beta}\delta g^{1\nu\beta}\delta g^{2\mu\alpha} + 6\delta g^{2\mu\alpha}\delta g^{2\nu}_{\alpha} - 12\delta g^{1\alpha\beta}\delta g^{1\mu\alpha}\delta g^{2\nu\beta} + 4\delta g^{1\nu}_{\alpha}\delta g^{3\mu\alpha} + 4\delta g^{1\mu\alpha}\delta g^{3\nu}_{\alpha} - \delta g^{4\mu\nu}
 \end{aligned}$$

Perturbation of the Christoffel symbols

General procedure

To calculate the perturbation of the Christoffel symbols, we first expand them into derivatives of the metric, before we apply the perturbation.

```
In[ ]:= ChristoffelCD[μ, -ρ, -ν]
ChristoffelToGradMetric[%]
```

Out[]=

$$\Gamma[\nabla]_{\rho\nu}^{\mu}$$

Out[]=

$$\frac{1}{2}g^{\mu\alpha}(-\partial_{\alpha}g_{\rho\nu} + \partial_{\nu}g_{\rho\alpha} + \partial_{\rho}g_{\nu\alpha})$$

First order

We have previously derived that the first order perturbation of the Christoffel symbols is a tensor field, which can be obtained from the covariant derivatives of the metric perturbation.

```
In[ ]:= Perturbation[ChristoffelCD[μ, -ρ, -ν], 1]
ExpandPerturbation[%];
SeparateMetric[ ][%]
```

Out[]=

$$\Delta[\Gamma[\nabla]_{\rho\nu}^{\mu}]$$

Out[]=

$$\frac{1}{2}\left(-g^{\mu\alpha}(\nabla_{\alpha}\delta g^1_{\rho\nu}) + g^{\mu\beta}(\nabla_{\nu}\delta g^1_{\beta\rho}) + g^{\mu\gamma}(\nabla_{\rho}\delta g^1_{\gamma\nu})\right)$$

Second order

We can use the same strategy to calculate the second order perturbation. First, we expand the Christoffel symbols into metric derivatives, and then perform the perturbative expansion.

```

In[ ]:= ChristoffelCD[μ, -ρ, -ν]
ChristoffelToGradMetric[%]
Perturbation[%, 2];
ExpandPerturbation[%]
ChangeCovD[%, PD, CD];
Expand[%];
ContractMetric[%];
ToCanonical[%];
Simplify[%]
SeparateMetric[ ][%]

```

Out[]:=

$$\Gamma[\nabla]^\mu_{\rho\nu}$$

Out[]:=

$$\frac{1}{2} g^{\mu\alpha} (-\partial_\alpha g_{\rho\nu} + \partial_\nu g_{\rho\alpha} + \partial_\rho g_{\nu\alpha})$$

Out[]:=

$$\frac{1}{2} \left((2 \delta g^1_{\beta\alpha} \delta g^{1\mu\beta} - \delta g^{2\mu\alpha}) (-\partial_\alpha g_{\rho\nu} + \partial_\nu g_{\rho\alpha} + \partial_\rho g_{\nu\alpha}) - 2 \delta g^{1\mu\alpha} (-\partial_\alpha \delta g^1_{\rho\nu} + \partial_\nu \delta g^1_{\rho\alpha} + \partial_\rho \delta g^1_{\nu\alpha}) + g^{\mu\alpha} (-\partial_\alpha \delta g^2_{\rho\nu} + \partial_\nu \delta g^2_{\rho\alpha} + \partial_\rho \delta g^2_{\nu\alpha}) \right)$$

Out[]:=

$$\frac{1}{2} \left(-(\nabla^\mu \delta g^2_{\nu\rho}) + \nabla_\nu \delta g^{2\mu}_\rho + 2 \delta g^{1\mu\alpha} (\nabla_\alpha \delta g^1_{\nu\rho} - \nabla_\nu \delta g^1_{\rho\alpha} - \nabla_\rho \delta g^1_{\nu\alpha}) + \nabla_\rho \delta g^{2\mu}_\nu \right)$$

Out[]:=

$$\frac{1}{2} \left(-g^{\mu\beta} (\nabla_\beta \delta g^2_{\nu\rho}) + g^{\mu\nu} (\nabla_\nu \delta g^2_{\gamma\rho}) + 2 g^{\alpha\kappa} g^{\mu\zeta} \delta g^1_{\zeta\kappa} (\nabla_\alpha \delta g^1_{\nu\rho} - \nabla_\nu \delta g^1_{\rho\alpha} - \nabla_\rho \delta g^1_{\nu\alpha}) + g^{\mu\xi} (\nabla_\rho \delta g^2_{\xi\nu}) \right)$$

Hence, also the second order perturbation of the Christoffel symbols is given as the covariant derivative of the metric perturbation, and thus also a tensor. Again we check our result:

```

In[ ]:= Perturbation[ChristoffelCD[μ, -ρ, -ν], 2]
ExpandPerturbation[%];
SeparateMetric[ ][%]

```

Out[]:=

$$\Delta^0[\Gamma[\nabla]^\mu_{\rho\nu}]$$

Out[]:=

$$-g^{\alpha\nu} g^{\mu\beta} \delta g^1_{\beta\nu} \left(-(\nabla_\alpha \delta g^1_{\rho\nu}) + \nabla_\nu \delta g^1_{\alpha\rho} + \nabla_\rho \delta g^1_{\alpha\nu} \right) + \frac{1}{2} \left(-g^{\mu\alpha} (\nabla_\alpha \delta g^2_{\rho\nu}) + g^{\mu\beta} (\nabla_\nu \delta g^2_{\beta\rho}) + g^{\mu\nu} (\nabla_\rho \delta g^2_{\gamma\nu}) \right)$$

Higher orders

We finally remark that also higher order perturbations are tensor fields, which are constructed by the same principle. Here we just show a few examples. A general formula can be derived in the same way as for the inverse metric.

```
In[ ]:= Perturbation[ChristoffelCD[μ, -ρ, -ν], 3]
ExpandPerturbation[%];
Expand[%];
ToCanonical[%];
Simplify[%]
```

Out[]:= $\Delta^1[\Gamma[\nabla]^\mu_{\rho\nu}]$

Out[]:=
$$\frac{1}{2} \left(-(\nabla^\mu \delta g^3_{\nu\rho}) + \nabla_\nu \delta g^{3\mu}_\rho + 3 \delta g^{2\mu\alpha} (\nabla_\alpha \delta g^1_{\nu\rho} - \nabla_\nu \delta g^1_{\rho\alpha} - \nabla_\rho \delta g^1_{\nu\alpha}) + 3 \delta g^{1\mu\alpha} (\nabla_\alpha \delta g^2_{\nu\rho} - \nabla_\nu \delta g^2_{\rho\alpha} + 2 \delta g^1_\alpha{}^\beta (-\nabla_\beta \delta g^1_{\nu\rho}) + \nabla_\nu \delta g^1_{\rho\beta} + \nabla_\rho \delta g^1_{\nu\beta}) - \nabla_\rho \delta g^2_{\nu\alpha} \right) + \nabla_\rho \delta g^{3\mu}_\nu$$

```
In[ ]:= Perturbation[ChristoffelCD[μ, -ρ, -ν], 4]
ExpandPerturbation[%];
Expand[%];
ToCanonical[%];
Simplify[%]
```

Out[]:= $\Delta^2[\Gamma[\nabla]^\mu_{\rho\nu}]$

Out[]:=
$$\frac{1}{2} \left(4 \delta g^{1\mu\alpha} (\nabla_\alpha \delta g^3_{\nu\rho}) - 12 \delta g^{1\mu\alpha} \delta g^2_\alpha{}^\beta (\nabla_\beta \delta g^1_{\nu\rho}) - 12 \delta g^1_\alpha{}^\beta \delta g^{1\mu\alpha} (\nabla_\beta \delta g^2_{\nu\rho}) + 24 \delta g^1_\alpha{}^\beta \delta g^1_\beta{}^\gamma \delta g^{1\mu\alpha} (\nabla_\gamma \delta g^1_{\nu\rho}) - \nabla^\mu \delta g^4_{\nu\rho} + 12 \delta g^{1\mu\alpha} \delta g^2_\alpha{}^\beta (\nabla_\nu \delta g^1_{\rho\beta}) - \right.$$

$$24 \delta g^1_\alpha{}^\beta \delta g^1_\beta{}^\gamma \delta g^{1\mu\alpha} (\nabla_\nu \delta g^1_{\rho\gamma}) + 12 \delta g^1_\alpha{}^\beta \delta g^{1\mu\alpha} (\nabla_\nu \delta g^2_{\rho\beta}) - 4 \delta g^{1\mu\alpha} (\nabla_\nu \delta g^3_{\rho\alpha}) + \nabla_\nu \delta g^{4\mu}_\rho + 4 \delta g^{3\mu\alpha} (\nabla_\alpha \delta g^1_{\nu\rho} - \nabla_\nu \delta g^1_{\rho\alpha} - \nabla_\rho \delta g^1_{\nu\alpha}) + 12 \delta g^{1\mu\alpha} \delta g^2_\alpha{}^\beta (\nabla_\rho \delta g^1_{\nu\beta}) -$$

$$\left. 24 \delta g^1_\alpha{}^\beta \delta g^1_\beta{}^\gamma \delta g^{1\mu\alpha} (\nabla_\rho \delta g^1_{\nu\gamma}) + 6 \delta g^{2\mu\alpha} (\nabla_\alpha \delta g^2_{\nu\rho} - \nabla_\nu \delta g^2_{\rho\alpha} + 2 \delta g^1_\alpha{}^\beta (-\nabla_\beta \delta g^1_{\nu\rho}) + \nabla_\nu \delta g^1_{\rho\beta} + \nabla_\rho \delta g^1_{\nu\beta}) - \nabla_\rho \delta g^2_{\nu\alpha} \right) + 12 \delta g^1_\alpha{}^\beta \delta g^{1\mu\alpha} (\nabla_\rho \delta g^2_{\nu\beta}) - 4 \delta g^{1\mu\alpha} (\nabla_\rho \delta g^3_{\nu\alpha}) + \nabla_\rho \delta g^{4\mu}_\nu$$

Perturbation of the Riemann curvature tensor

General procedure

Next we expand the Riemann curvature tensor. Here we aim to express it only through the perturbation of the Christoffel symbols, but not break it further down to the metric, which would be lengthy. We thus make use of its definition.

```
In[ ]:= RiemannCD[-σ, -ρ, -ν, μ]
ChangeCurvature[%]
```

Out[]:= $R[\nabla]_{\sigma\rho\nu}{}^\mu$

Out[]:=
$$\Gamma[\nabla]^\alpha_{\sigma\nu} \Gamma[\nabla]^\mu_{\rho\alpha} - \Gamma[\nabla]^\alpha_{\rho\nu} \Gamma[\nabla]^\mu_{\sigma\alpha} + \partial_\rho \Gamma[\nabla]^\mu_{\sigma\nu} - \partial_\sigma \Gamma[\nabla]^\mu_{\rho\nu}$$

First order

We can use this formula to calculate its perturbation. Now we use our previously gained knowledge that the perturbation of the Christoffel symbol is a tensor. We can thus combine partial derivatives and Christoffel symbols to covariant derivatives to get the desired result:

```
In[ ]:= RiemannCD[-σ, -ρ, -ν, μ];
ChangeCurvature[%];
Perturbation[%, 1]
ChangeCovD[%, PD, CD];
ToCanonical[%]
```

```
Out[ ]:=
- Γ[ν]μσ α Δ[Γ[ν]αρ ν] + Γ[ν]μρ α Δ[Γ[ν]ασ ν] + Γ[ν]ασ ν Δ[Γ[ν]μρ α] - Γ[ν]αρ ν Δ[Γ[ν]μσ α] + ∂ρ Δ[Γ[ν]μσ ν] - ∂σ Δ[Γ[ν]μρ ν]
```

```
Out[ ]:=
∇ρ Δ[Γ[ν]μν σ] - ∇σ Δ[Γ[ν]μν ρ]
```

Second order

We continue with the second order. We see that we obtain the same structure. We find a covariant derivative of the perturbation of the Christoffel symbols, and products of lower order perturbations from the quadratic terms.

```
In[ ]:= RiemannCD[-σ, -ρ, -ν, μ];
ChangeCurvature[%];
Perturbation[%, 2];
ChangeCovD[%, PD, CD];
ToCanonical[%, UseMetricOnVBundLe → None]
```

```
Out[ ]:=
2 Δ[Γ[ν]αν σ] Δ[Γ[ν]μρ α] - 2 Δ[Γ[ν]αν ρ] Δ[Γ[ν]μσ α] + ∇ρ Δ0[Γ[ν]μν σ] - ∇σ Δ0[Γ[ν]μν ρ]
```

Higher orders

To study higher orders, it is instructive to first expand a few terms using the method above and look at their structure.

```
In[ ]:= Table[Perturbation[ChangeCurvature[RiemannCD[-σ, -ρ, -ν, μ]], k], {k, 4}];
ChangeCovD[#, PD, CD] & /@ %;
ToCanonical[#, UseMetricOnVBundLe → None] & /@ %;
ScreenDollarIndices /@ %;
TableForm[%]
```

```
Out[ ]//TableForm=
∇ρ Δ[Γ[ν]μν σ] - ∇σ Δ[Γ[ν]μν ρ]
2 Δ[Γ[ν]αν σ] Δ[Γ[ν]μρ α] - 2 Δ[Γ[ν]αν ρ] Δ[Γ[ν]μσ α] + ∇ρ Δ0[Γ[ν]μν σ] - ∇σ Δ0[Γ[ν]μν ρ]
- 3 Δ[Γ[ν]μσ α] Δ0[Γ[ν]αν ρ] + 3 Δ[Γ[ν]μρ α] Δ0[Γ[ν]αν σ] + 3 Δ[Γ[ν]αν σ] Δ0[Γ[ν]μρ α] - 3 Δ[Γ[ν]αν ρ] Δ0[Γ[ν]μσ α] + ∇ρ Δ1[Γ[ν]μν σ] - ∇σ Δ1[Γ[ν]μν ρ]
- 4 Δ[Γ[ν]μσ α] Δ1[Γ[ν]αν ρ] + 4 Δ[Γ[ν]μρ α] Δ1[Γ[ν]αν σ] + 6 Δ0[Γ[ν]αν σ] Δ0[Γ[ν]μρ α] + 4 Δ[Γ[ν]αν σ] Δ1[Γ[ν]μρ α] - 6 Δ0[Γ[ν]αν ρ] Δ0[Γ[ν]μσ α] - 4 Δ[Γ[ν]αν ρ] Δ1[Γ[ν]μσ α] + ∇ρ Δ2[Γ[ν]μν σ] - ∇σ Δ2[Γ[ν]μν ρ]
```

Now it is indeed not difficult to guess a general formula.

```

In[ ]:= Table[CD[-ρ][Perturbation[ChristoffelCD[μ, -σ, -ν], k]] - CD[-σ][Perturbation[ChristoffelCD[μ, -ρ, -ν], k]] +
  Sum[Binomial[k, j](Perturbation[ChristoffelCD[μ, -ρ, -α], j] × Perturbation[ChristoffelCD[α, -σ, -ν], k - j] - Perturbation[ChristoffelCD[μ, -σ, -α], j] × Perturbation[ChristoffelCD[α, -ρ, -ν], k - j]),
  {j, k - 1}], {k, 4};
ToCanonical[#, UseMetricOnVBundLe → None] & /@ %;
ScreenDollarIndices /@ %;
TableForm[%]

```

Out[]//TableForm=

$$\begin{aligned}
& \nabla_{\rho} \Delta[\Gamma[\nabla]_{\nu\sigma}^{\mu}] - \nabla_{\sigma} \Delta[\Gamma[\nabla]_{\nu\rho}^{\mu}] \\
& 2 \Delta[\Gamma[\nabla]_{\nu\sigma}^{\alpha}] \Delta[\Gamma[\nabla]_{\rho\alpha}^{\mu}] - 2 \Delta[\Gamma[\nabla]_{\nu\rho}^{\alpha}] \Delta[\Gamma[\nabla]_{\sigma\alpha}^{\mu}] + \nabla_{\rho} \Delta^0[\Gamma[\nabla]_{\nu\sigma}^{\mu}] - \nabla_{\sigma} \Delta^0[\Gamma[\nabla]_{\nu\rho}^{\mu}] \\
& -3 \Delta[\Gamma[\nabla]_{\sigma\alpha}^{\mu}] \Delta^0[\Gamma[\nabla]_{\nu\rho}^{\alpha}] + 3 \Delta[\Gamma[\nabla]_{\rho\alpha}^{\mu}] \Delta^0[\Gamma[\nabla]_{\nu\sigma}^{\alpha}] + 3 \Delta[\Gamma[\nabla]_{\nu\sigma}^{\alpha}] \Delta^0[\Gamma[\nabla]_{\rho\alpha}^{\mu}] - 3 \Delta[\Gamma[\nabla]_{\nu\rho}^{\alpha}] \Delta^0[\Gamma[\nabla]_{\sigma\alpha}^{\mu}] + \nabla_{\rho} \Delta^1[\Gamma[\nabla]_{\nu\sigma}^{\mu}] - \nabla_{\sigma} \Delta^1[\Gamma[\nabla]_{\nu\rho}^{\mu}] \\
& -4 \Delta[\Gamma[\nabla]_{\sigma\alpha}^{\mu}] \Delta^1[\Gamma[\nabla]_{\nu\rho}^{\alpha}] + 4 \Delta[\Gamma[\nabla]_{\rho\alpha}^{\mu}] \Delta^1[\Gamma[\nabla]_{\nu\sigma}^{\alpha}] + 6 \Delta^0[\Gamma[\nabla]_{\nu\sigma}^{\alpha}] \Delta^0[\Gamma[\nabla]_{\rho\alpha}^{\mu}] + 4 \Delta[\Gamma[\nabla]_{\nu\sigma}^{\alpha}] \Delta^1[\Gamma[\nabla]_{\rho\alpha}^{\mu}] - 6 \Delta^0[\Gamma[\nabla]_{\nu\rho}^{\alpha}] \Delta^0[\Gamma[\nabla]_{\sigma\alpha}^{\mu}] - 4 \Delta[\Gamma[\nabla]_{\nu\rho}^{\alpha}] \Delta^1[\Gamma[\nabla]_{\sigma\alpha}^{\mu}] + \nabla_{\rho} \Delta^2[\Gamma[\nabla]_{\nu\sigma}^{\mu}] - \nabla_{\sigma} \Delta^2[\Gamma[\nabla]_{\nu\rho}^{\mu}]
\end{aligned}$$